
	Reason for change:
	Originally IP handling has been restricted to EUTRA; therefore a "ChangeCell" command was introduced to change the cell id for all entries of routing table independent from the DRB id.

This has been enhanced to allow changing to another RAT by changing the DrbInfo for all entries of the routing table to a specific radio bearer. This serves the purpose for one-PDN-only scenarios but in general will not work for multi-PDN IRAT scenarios which need to be considered in context of SRVCC (e.g. R5-152145) and WLAN interworking.

Use cases to be supported in TTCN:

1. Intra EUTRA cell change: 
EUTRA cell id to be changed for all entries of the routing table as required EUTRA HO scenarios. 
NOTE: This is not an issue for UTRAN as here utran_CellDedicated is used for all entries.

2. IRAT scenarios:
DrbInfo to be modified per PDN.



	
	

	Summary of change:
	· Separated functions to change routing table acc. particular use case
· Common table at IP_PTC for mapping of PDNs to DRBs 
( no DrbInfo needs to be explicitly stored for IMS or address servers anymore
· Distinct commands for coordination messages acc. particular use case

	
	

	Consequences if not approved:

· Non future proven implementation (e.g. regarding multi-PDN IRAT scenarios)

	Other comments:
	This CR is related to R5s150258 but provides a common concept to support multi-PDN scenarios


Change 1 – Routing table management
	Function name
	f_IP_DrbInfo_ChangeCell (removed)
f_RoutingTable_ChangeCell (removed)
f_IP_DrbInfo_DrbIdMatch (new)
f_RoutingTable_ChangeDrbInfo (new)
f_IP_DrbInfo_ChangeEutraCell (new)
f_RoutingTable_ChangeEutraCell (new)

	Reason for change
	There are distinct use cases to change the DrbInfo of routing table entries:
1. to change the the cellId for EUTRA DRBs
NOTE: there is no such use case for UTRAN or GERAN

2. to change the whole DrbInfo (e.g. for IRAT HO); in general this requires to consider different PDNs i.e. the DrbInfo shall be changed for a given (old) DrbInfo only.

	Summary of changes
	New functions fulfilling the above requirements; 
(old functions removed)

	TTCN module
	IP_PTC_Routing


	

















	





















	  function f_IP_DrbInfo_DrbIdMatch(IP_DrbInfo_Type p_DrbInfo1,

                                   IP_DrbInfo_Type p_DrbInfo2) return boolean 

  { /* returns true when both DrbInfo parameters refer to the same DrbId;

       this implies that the DRB belongs to the same RAT; the CellId is not checked */

    if (ischosen(p_DrbInfo1.Eutra) and ischosen(p_DrbInfo2.Eutra)) {

      if (p_DrbInfo1.Eutra.DrbId == p_DrbInfo2.Eutra.DrbId) {

        return true;

      }

    }

    else if (ischosen(p_DrbInfo1.Utran) and ischosen(p_DrbInfo2.Utran)) {

      if (p_DrbInfo1.Utran.DrbId == p_DrbInfo2.Utran.DrbId) {

        return true;

      }

    }

    else if (ischosen(p_DrbInfo1.Geran) and ischosen(p_DrbInfo2.Geran)) {

      if (p_DrbInfo1.Geran.DrbId == p_DrbInfo2.Geran.DrbId) {

        return true;

      }

    }

    return false;

  }

	  function f_RoutingTable_ChangeDrbInfo(inout IP_RoutingTable_Type p_RoutingTable,

                                        IP_DrbInfo_Type p_DrbInfoOld,

                                        IP_DrbInfo_Type p_DrbInfoNew) return boolean 

  { /* for all entries: if DrbInfo matches p_DrbInfoOld => DrbInfo gets changed to p_DrbInfoNew;

       returns true when at least one entry has been changed */

    var boolean v_TableIsChanged := false;

    var integer i;

    for (i := 0; i < lengthof(p_RoutingTable); i := i+1) {

      if (f_IP_DrbInfo_DrbIdMatch(p_RoutingTable[i].DRB, p_DrbInfoOld)) {

        p_RoutingTable[i].DRB := p_DrbInfoNew;

        v_TableIsChanged := true;

      }

    }

    return v_TableIsChanged;

  }

	  function f_IP_DrbInfo_ChangeEutraCell(inout IP_DrbInfo_Type p_DrbInfo,

                                        EUTRA_CellId_Type p_EutraCellId) return boolean 

  { /* for given DrbInfo: if the DrbInfo refers to EUTRA the cellId gets modified;

       returns true when p_DrbInfo has been changed */

    var boolean v_IsChanged := false;

    if (ischosen(p_DrbInfo.Eutra)) {

      if (p_DrbInfo.Eutra.CellId != p_EutraCellId) {

        p_DrbInfo.Eutra.CellId := p_EutraCellId;

        v_IsChanged := true;

      }

    }

    return v_IsChanged;

  }

	  function f_RoutingTable_ChangeEutraCell(inout IP_RoutingTable_Type p_RoutingTable,

                                          EUTRA_CellId_Type p_EutraCellId) return boolean

  { /* for all entries: modify cellId if DrbInfo refers to EUTRA;

       returns true when at least one entry has been changed */

    var boolean v_TableIsChanged := false;

    var integer i;

    for (i := 0; i < lengthof(p_RoutingTable); i := i+1) {

      v_TableIsChanged := v_TableIsChanged or f_IP_DrbInfo_ChangeEutraCell(p_RoutingTable[i].DRB, p_EutraCellId);

    }

    return v_TableIsChanged;

  }


Change 2 – IP_PTC
Change 2.1 – New definitions
	Type/function name
	PDN_MappingInfo_Type (new), 
PDN_MappingTable_Type  (new), 
f_IP_PTC_PDN_SetDrbInfoForDefaultDrb (new),
f_IP_PTC_PDN_GetDrbInfoForDefaultDrb (new),
IP_PTC, 
fl_IP_PTC_Init

	Reason for change
	Mapping of DrbInfo to PDNs is global for the PDN (i.e. all network entities of one PDN shall be mapped to the same DRB)

	Summary of changes
	New tabele vc_IP_PTC_PDN_MappingTable; associated access function

	TTCN module
	IP_PTC_Component, IP_PTC_Main


	  type record PDN_MappingInfo_Type {

    IP_DrbInfo_Type DefaultDrb optional                         /* omit => no IP handling for the respective PDN */     

  };

	  type record of PDN_MappingInfo_Type PDN_MappingTable_Type;

	  type component IP_PTC {
    var IP_RoutingTable_Type    vc_IP_PTC_DRB_RoutingTable;

    var PDN_IP_AddrServer_Type  vc_IP_PTC_PDN_AddrServer;

    var IP_PTC_IMS_Info_Type    vc_IP_PTC_IMS_Info;
    var PDN_MappingTable_Type   vc_IP_PTC_PDN_MappingTable;
    // ports connected to IMS:

    port IP_IMS_CTRL_PORT       IMS_CTRL[tsc_NoOfPDNs];

    port IP_IMS_SERVER_PORT     IMS_Server[tsc_NoOfPDNs];

    port IP_IMS_CLIENT_PORT     IMS_Client[tsc_NoOfPDNs];

    // coordination ports:

    port IP_PTC_CTRL_PORT       EUTRA_CTRL;

    port IP_PTC_CTRL_PORT       UTRAN_CTRL;

    port IP_PTC_CTRL_PORT       GERAN_CTRL;

    // coordination ports to HTTP_PTC

    port IP_HTTP_CTRL_PORT      HTTPCTRL;

    port IP_HTTP_DATA_PORT      HTTPDATA;

    // system ports

    port IP_CONTROL_PORT        IP_CTRL;

    port IPSEC_CONTROL_PORT     IPSEC_CTRL;

    port IP_SOCKET_PORT         IP_SOCK;

  };

	  function f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(PDN_Index_Type p_PdnIndex,

                                                IP_DrbInfo_Type p_DrbInfo) runs on IP_PTC

  {

    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);

    vc_IP_PTC_PDN_MappingTable[v_Index].DefaultDrb := p_DrbInfo;

  }

	  function f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(PDN_Index_Type p_PdnIndex) runs on IP_PTC return IP_DrbInfo_Type

  {

    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);

    return vc_IP_PTC_PDN_MappingTable[v_Index].DefaultDrb;

  }

	  function fl_IP_PTC_Init() runs on IP_PTC

  {

    var integer i;

    vc_IP_PTC_DRB_RoutingTable := {};

    for (i:=0; i < tsc_NoOfPDNs; i:=i+1) {

      vc_IP_PTC_PDN_AddrServer.DHCP[i]        := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4[i] := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6[i] := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_PDN_AddrServer.ICMPv6[i]      := valueof(cs_IP_AddrServer_Init);

      vc_IP_PTC_IMS_Info.Server[i]            := valueof(cs_IMS_Server_Init);
      vc_IP_PTC_PDN_MappingTable[i].DefaultDrb := omit;
    }

  }


Change 2.2 – Changed definitions
	Type name
	AddrServerInfo_Type, 
IMS_Server_Type

	Reason for change
	DRB mapping shall not be specific for network entities but can be handled globally depending on the PDN the entity belongs to.

	Summary of changes
	DrbInfo removed from AddrServerInfo_Type and IMS_Server_Type; PdnIndex added to IMS_Server_Type

	TTCN module
	IP_PTC_Component


	  type record AddrServerInfo_Type {             /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, UTRAN) */

    charstring         ServerAddr       optional,               /* the server's address;

                                                                   global address in case of IPv6 */

    charstring         ClientAddr       optional,               /* IP address to be assigned to the UE via DHCP or ICMPv6;

                                                                   global address in case of IPv6 */



    PDN_Index_Type     PdnIndex         optional                /* @sic R5-145754 sic@ */

  };

	  type record IMS_Server_Type {                                 /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, UTRAN) */

    IMS_SecurityContext_Type            UnprotectedContext,

    IMS_SecurityContext_Type            ProtectedContext1   optional,

    IMS_SecurityContext_Type            ProtectedContext2   optional,

    IP_AddrInfo_Type                    IPv4Addr            optional,   /* the server's IPv4 address, omit as long as the server is not started */

    IP_AddrInfo_Type                    IPv6Addr            optional,   /* the server's IPv6 address, omit as long as the server is not started */

    IPsec_SecurityKeys_Type             SecurityKeys        optional,   /* security keys to be applied for security associations */

    IMS_RegistrationAddress_Type        RegistrationAddress optional,   /* if present => UE has at least started the registration and in case of non-GIBA

                                                                           IPsec and UDP server for protected signalling are started */

    PDN_Index_Type                      PdnIndex            optional    /* omit as long as the server is not started */




  };


Change 3 – IMS handling
	Type/function name
	f_ImsServer_IsStarted, 
fl_IMS_Server_AddSecurityContext, 
f_IMS_Server_ChangeCell (removed), 
cs_IMS_Server_Init, 
f_IP_IMS_Start, 
fl_IP_IMS_Server_SendDL, 
fl_IP_ImsServer_Start

	Reason for change
	Acc. to change 2 the IMS server does not store any DrbInfo anymore but a PdnIndex instead

	Summary of changes
	Whenever the IMS server needs the DrbInfo it shall retrieve it from the global PDN mapping

	TTCN module
	IP_PTC_IMS_Handler, IP_PTC_Main


	  function f_ImsServer_IsStarted(IMS_Server_Type p_ImsServer) return boolean

  {

    return isvalue(p_ImsServer.PdnIndex);

  }

	  function fl_IMS_Server_AddSecurityContext(inout IMS_Server_Type p_ImsServer,

                                            IMS_RegistrationInfo_Type p_RegistrationInfo) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var IP_AddrInfo_Type v_LocalIpAddr := p_RegistrationInfo.NW_Address;

    var IP_AddrInfo_Type v_RemoteIpAddr := p_RegistrationInfo.UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo := p_RegistrationInfo.SecurityInfo;

    var IMS_ProtectedPorts_Type v_ProtectedPorts := v_SecurityInfo.ProtectedPorts;

    var PortNumber_Type v_LocalPortServer := v_ProtectedPorts.Port_ps;

    var PortNumber_Type v_LocalPortClient := v_ProtectedPorts.Port_pc;

    var IMS_SecurityContext_Type v_SecurityContext := valueof(cs_IMS_SecurityContext(cs_IMS_SecurityContextInfo_Protected(v_SecurityInfo)));


    var IP_DrbInfo_Type v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(p_ImsServer.PdnIndex);
    var boolean v_IsSecondContext := false;

    if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

      if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {

        FatalError(__FILE__, __LINE__, "Not more than two security contexts supported at the same time");

      }

      if (v_LocalPortServer != p_ImsServer.ProtectedContext1.SecurityInfo.Protected.ProtectedPorts.Port_ps) {

        FatalError(__FILE__, __LINE__, "re-configuration of TCP/UDP server not supported");

        /* NOTE: if the server port changes during re-authentication we would need to have 2 TCP and 2 UDP servers in parallel until re-authentication has finished */

      }

      if (v_LocalPortClient == p_ImsServer.ProtectedContext1.SecurityInfo.Protected.ProtectedPorts.Port_pc) {

        FatalError(__FILE__, __LINE__, "re-authentication: client port of new security context shall differ from the old one");    /* in general a new port shall be assigned at the IMS PTC by means of f_IMS_PTC_Security_Init */

      }

      p_ImsServer.ProtectedContext2 := v_SecurityContext;

      v_IsSecondContext := true;

    } else {

      p_ImsServer.ProtectedContext1 := v_SecurityContext;

    }

    // get SPIs for the network side:

    // v_SecurityInfo.SPIs := f_SPIs_SetNetworkValues(v_SPIs);  // @sic R5s130266 change 1.3, 4, 5 - MCC160 Implementation: SPIs are set at the IMS PTC now sic@ */

    fl_IPsec_Configure(v_LocalIpAddr, v_RemoteIpAddr, v_SecurityInfo, p_ImsServer.SecurityKeys);

    if (not v_IsSecondContext) {     /* servers need to be started only for the first context */

      f_UdpServer_Start(v_LocalIpAddr, v_LocalPortServer, v_DrbInfo);

      f_TcpServer_Start(v_LocalIpAddr, v_LocalPortServer, v_DrbInfo);

    }

    f_UdpClient_Start(v_LocalIpAddr, v_LocalPortClient, v_DrbInfo);  /* @sic R5s130113 change 4 sic@ */

  }

	







	  template (value) IMS_Server_Type cs_IMS_Server_Init :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, UTRAN) */

    UnprotectedContext := cs_IMS_SecurityContext(cs_IMS_SecurityContextInfo_Unprotected),  /* @sic R5-145732: support of 2nd security context sic@ */

    ProtectedContext1 := omit,                                                             /* @sic R5-145732: support of 2nd security context sic@ */

    ProtectedContext2 := omit,                                                             /* @sic R5-145732: support of 2nd security context sic@ */

    IPv4Addr := omit,

    IPv6Addr := omit,

    SecurityKeys := omit,

    RegistrationAddress := omit,                                                           /* @sic R5-145732: support of 2nd security context sic@ */

    PdnIndex:= omit

  };

	  function f_IP_IMS_Start(IMS_Server_Type p_ImsServer,

                          charstring p_P_CSCF_IPv4Address,

                          charstring p_P_CSCF_IPv6Address,
                          PDN_Index_Type p_PdnIndex
) runs on IP_PTC return IMS_Server_Type

  {

    var IMS_Server_Type v_ImsServer := p_ImsServer;

    var PortNumber_Type v_PortNumber := v_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_ps;
    var IP_DrbInfo_Type v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(p_PdnIndex);
    v_ImsServer.IPv4Addr := valueof(cs_IPv4Addr(p_P_CSCF_IPv4Address));

    v_ImsServer.IPv6Addr := valueof(cs_IPv6Addr(p_P_CSCF_IPv6Address));


    v_ImsServer.PdnIndex := p_PdnIndex;
    /* Start UDP servers for the given PDN;

     * Note: if there is already a UDP server using the same IP address and port, f_UdpServer_Start will cause an error raised by the SS;

     *       => it is implicitly checked that different PDNs have different addresses as requested by 36.523-3 cl. 4.2.4.6 */

    f_UdpServer_Start(v_ImsServer.IPv4Addr, v_PortNumber, v_DrbInfo);

    f_UdpServer_Start(v_ImsServer.IPv6Addr, v_PortNumber, v_DrbInfo);

    /* Start TCP servers for the given PDN */

    f_TcpServer_Start(v_ImsServer.IPv4Addr, v_PortNumber, v_DrbInfo);

    f_TcpServer_Start(v_ImsServer.IPv6Addr, v_PortNumber, v_DrbInfo);

    return v_ImsServer;

  }

	  function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,

                                   IMS_RoutingInfo_Type p_RoutingInfo,

                                   IMS_RequestResponse_Type p_RequestResponse,

                                   bitstring p_Data) runs on IP_PTC

  { /* @sic R5-145732: support of 2nd security context sic@ */

    var InternetProtocol_Type v_Protocol := p_RoutingInfo.Protocol;

    var IMS_SecurityContextEnum_Type v_SecurityContextEnum;

    var IMS_SecurityContext_Type v_SecurityContext;

    var IMS_LocalRemotePort_Type v_LocalRemotePort;

    var template (omit) PortNumber_Type v_LocalPort;

    var PortNumber_Type v_RemotePort;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var template (value) IP_Socket_Type v_LocalSocket;

    var template (value) IP_Socket_Type v_RemoteSocket;

    var template (omit) IP_Connection_Type v_TCP_Connection := omit;

    var IP_Connection_Type v_NewTcpConnectionByRef;

    var Datagram_Content_Type v_Data := bit2oct(p_Data);

    var boolean v_IsUnprotected;

    var IP_DrbInfo_Type v_DrbInfo;
    if (isvalue(p_RoutingInfo.Security)) {

      /* Security may be omitted by the IMS PTC when the IMS message shall be sent protected or unprotected based on whether security is configured or not;

         in case of the initial REGISTER response security is already configured but the reponse needs to be sent unprotected (as the UE has not fully configured security yet)

         => at least in this case the IMS PTC shall set the "Security" field of the RoutingInfo */

      v_SecurityContextEnum := p_RoutingInfo.Security;

    } else {

      if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {         /* the latest security context is preferred */

        v_SecurityContextEnum := protectedContext2;

      } else if (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) {

        v_SecurityContextEnum := protectedContext1;

      } else {

        v_SecurityContextEnum := unprotected;

      }

    }

    if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, v_SecurityContextEnum)) {

      FatalError(__FILE__, __LINE__, "IMS security not established");

    }

    v_SecurityContext := valueof(fl_IMS_Server_GetSecurityContext(p_ImsServer, v_SecurityContextEnum));

    v_IsUnprotected := (v_SecurityContextEnum == unprotected);

    v_LocalRemotePort := fl_IP_IMS_GetPortsTX(v_SecurityContext, v_Protocol, p_RequestResponse);

    v_LocalPort  := v_LocalRemotePort.Local;

    v_RemotePort := v_LocalRemotePort.Remote;

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {

      v_NW_Address := p_RoutingInfo.NW_Address;

      v_UE_Address := p_RoutingInfo.UE_Address;

    } else if (fl_ImsServer_UeIsRegistered(p_ImsServer)) {

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

    } else {

      FatalError(__FILE__, __LINE__, "missing IP addresses");

    }

    select (v_Protocol) {

      case (udp) {

        v_LocalSocket  := cs_IP_Socket(v_NW_Address, v_LocalPort);

        v_RemoteSocket := cs_IP_Socket(v_UE_Address, v_RemotePort);

        IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_Data));

      }

      case (tcp) {

        select (p_RequestResponse) {

          case (response) {

            v_TCP_Connection := v_SecurityContext.TcpServer;

          }

          case (request) {

            v_TCP_Connection := v_SecurityContext.TcpClient;

            if (not isvalue(v_TCP_Connection)) {

              if (v_IsUnprotected) {

                v_LocalPort := omit;  // in case of an unprotected TCP connection to UE there is no need to specify any specific local port but we can use an ephemeral port instead

              }

              v_NewTcpConnectionByRef := f_TcpClient_ConnectReq(cs_IP_Socket(v_NW_Address, v_LocalPort), cs_IP_Socket(v_UE_Address, v_RemotePort), ims);

              v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(p_ImsServer.PdnIndex);
              alt {

                [] a_IP_IMS_TCP_ConnectionControl(p_ImsServer) {  /* @sic R5s140713 - MCC160 implementation sic@ */

                  repeat;

                }

                [] a_TcpClient_ConnectCnf(v_NewTcpConnectionByRef, v_DrbInfo) {

                  v_SecurityContext.TcpClient := v_NewTcpConnectionByRef;

                  fl_IMS_Server_SetSecurityContext(p_ImsServer, v_SecurityContextEnum, v_SecurityContext); // @sic R5s141322 change 2 sic@

                  v_TCP_Connection := v_NewTcpConnectionByRef;

                }

              }

            }

          }

        }

        if (not isvalue(v_TCP_Connection)) {

          /* @sic R5s150270 change 2: just skip the data sic@

             there can be a race condion when the UE does not wait for any 200 OK but immediately closes the TCP connection with the server

             => when TTCN wants to send the 200 OK the connection is gone already */

        } else {

          IP_SOCK.send(cs_TCP_DATA_REQ(valueof(v_TCP_Connection), v_Data));

        }

      }

      case else {

        FatalError(__FILE__, __LINE__, "invalid protocol");

        return;

      }

    }

  }

	  function fl_IP_ImsServer_Start(PDN_Index_Type p_PdnIndex
) runs on IP_PTC

  { /* Multiple PDNs

     * In case of multiple PDNs (e.g. for emergency calls) there can be

     * 1) multiple P-CSCFs i.e. one for each PDN; this is typically the case for roaming scenarios when the normal IMS registration is done at the P-CSCF in th home PLMN

     *    whereas the emergency registration is done in the visited PLMN

     * 2) one P-CSCF for (both) PDNs, e.g. when visited PLMN is the home PLMN

     *

     * Assumption:

     *   even in case of one P-CSCF for (both) PDNs there shall be different P-CSCF IP addresse i.e. the P-CSCF shall have a different IP address for each PDN.

     *   (in case of IPv6 as working assumption the P-CSCF and all other entities belonging to the PDN shall have the same network prefix and the network prefix for different PDNs shall be different;

     *   see TS 36.523-3 cl. 4.2.4.6)

     * => There can be different UDP servers for the PDNs and the routing table can be configured with IP address and port of the network side only

     *    (Note: if the P-CSCF would have only one IP address for two PDNs in DL the routing can only be determined based on the destination address

     *    i.e. the UE's addresses need to be considered when starting the UDP server - what is weird from socket programming point of view) */

    /* @sic R5s130362 - MCC160 Implementation: tsc_PDN_AddressInfo sic@ */

    var integer v_IndexVal := f_PDN_PdnIndex2Integer(p_PdnIndex);

    var IMS_Server_Type v_ImsServer := vc_IP_PTC_IMS_Info.Server[v_IndexVal];

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    v_ImsServer := f_IP_IMS_Stop(v_ImsServer);                        // @sic R5s120065 MCC160 Implementation; R5s150118 sic@

    vc_IP_PTC_IMS_Info.Server[v_IndexVal] := f_IP_IMS_Start(v_ImsServer,

                                                            v_PDN_AddressInfo.PCSCF_IPAddressIPv4,

                                                            v_PDN_AddressInfo.PCSCF_IPAddressIPv6,

                                                            p_PdnIndex);

  }


Change 4 – Handling of non-IMS servers 
	Type/function name
	fl_IP_AddrServer_GetDrbInfo, 
cs_IP_AddrServer_Init, 
fl_IP_AddrServer_IsRunning, 
fl_IP_AddrServer_Find, 
fl_IP_UdpServer_Start, 
fl_IP_AddrServers_Start, 
a_DNS_Handler

	Reason for change
	Acc. to change 2 the address servers do not store any DrbInfo anymore 

	Summary of changes
	Whenever a server needs the DrbInfo it shall retrieve it from the global PDN mapping;
f_IP_DrbInfo_DrbIdMatch can be reused in fl_IP_AddrServer_Find

	TTCN module
	IP_PTC_Main


	  function fl_IP_AddrServer_GetDrbInfo(IP_AddrServerArray_Type p_AddrServerArray,

                                       IP_AddrInfo_Type p_ServerAddrInfo) return template (omit) IP_DrbInfo_Type

  {

    var template (omit) AddrServerInfo_Type v_AddrServerInfo := fl_IP_AddrServer_Get(p_AddrServerArray, p_ServerAddrInfo);

    var template (omit) IP_DrbInfo_Type v_IP_DrbInfo := omit;

    if (isvalue(v_AddrServerInfo.PdnIndex)) {


      v_IP_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(valueof(v_AddrServerInfo.PdnIndex));
    }

    return v_IP_DrbInfo;

  }

	  template (value) AddrServerInfo_Type cs_IP_AddrServer_Init(template (omit) charstring p_ServerAddr := omit,

                                                             template (omit) charstring p_ClientAddr := omit,


                                                             template (omit) PDN_Index_Type p_PdnIndex := omit) :=

  { /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, UTRAN) */

    ServerAddr := p_ServerAddr,

    ClientAddr := p_ClientAddr,


    PdnIndex :=  p_PdnIndex              /* @sic R5-145754  sic@ */

  };

	  function fl_IP_AddrServer_IsRunning(AddrServerInfo_Type p_ServerInfo) return boolean

  {

    return isvalue(p_ServerInfo.PdnIndex);

  }

	function fl_IP_AddrServer_Find(IP_AddrServerArray_Type p_ServerArray,

                                 template (omit) IP_DrbInfo_Type p_DrbInfo) runs on IP_PTC return integer

  {

    var IP_DrbInfo_Type v_DrbInfoPdn;

    var IP_DrbInfo_Type v_DrbInfo;

    var integer i;

    if (isvalue(p_DrbInfo)) {

      v_DrbInfo := valueof(p_DrbInfo);

      for (i:=0; i < lengthof(p_ServerArray); i:=i+1) { 
        if (fl_IP_AddrServer_IsRunning(p_ServerArray[i])) {

          v_DrbInfoPdn := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(p_ServerArray[i].PdnIndex);

          if (f_IP_DrbInfo_DrbIdMatch(v_DrbInfoPdn, v_DrbInfo)) {

            return i;

          }

        }

















        }

      }

    }

    FatalError(__FILE__, __LINE__, "No Server found for given DRB");

    return -1;

  }

	  function fl_IP_UdpServer_Start(charstring p_ServerAddr,

                                 template (omit) charstring p_ClientAddr,

                                 integer p_ServerPort,

                                 IP_DrbInfo_Type p_DrbInfo,

                                 PDN_Index_Type p_PdnIndex) runs on IP_PTC return AddrServerInfo_Type

  {

    var template (value) IP_AddrInfo_Type v_IP_AddrInfo := f_Char2IpAddress(p_ServerAddr);  /* @sic R5s141324: IPv4 or IPv6 address sic@ */

    f_UdpServer_Start(v_IP_AddrInfo, p_ServerPort, p_DrbInfo);

    return valueof(cs_IP_AddrServer_Init(p_ServerAddr, p_ClientAddr, p_PdnIndex));

  }

	  function fl_IP_AddrServers_Start(PDN_Index_Type p_PdnIndex,

                                   IP_DrbInfo_Type p_DrbInfo) runs on IP_PTC

  { /* @sic R5-113734 change 8 sic@ */

    /* @sic R5s130362 - MCC160 Implementation: tsc_PDN_AddressInfo sic@ */

    var PDN_AddressInfo_Type v_PDN_AddressInfo := f_PDN_AddressInfo_Get(p_PdnIndex);

    var charstring v_ServerIpAddrv4 := v_PDN_AddressInfo.DHCP_ServerAddress;

    var charstring v_ServerIpAddrv6 := v_PDN_AddressInfo.ICMPv6_ServerAddress;

    var charstring v_ClientIpAddrv4 := v_PDN_AddressInfo.UE_IPAddressIPv4;

    var charstring v_ClientIpAddrv6 := v_PDN_AddressInfo.UE_IPAddressIPv6;

    var integer v_Index := f_PDN_PdnIndex2Integer(p_PdnIndex);

    fl_IP_AddrServers_Stop(v_Index);    // stop sever if any running

    // DHCP

    vc_IP_PTC_PDN_AddrServer.DHCP[v_Index] := fl_IP_UdpServer_Start(v_ServerIpAddrv4, v_ClientIpAddrv4, tsc_PortNumber_DHCP_Server, p_DrbInfo, p_PdnIndex);   // @sic R5-145754  sic@

    // DNS

    vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4[v_Index] := fl_IP_UdpServer_Start(v_ServerIpAddrv4, omit, tsc_PortNumber_DNS_Server, p_DrbInfo, p_PdnIndex);         // @sic R5-145754  sic@

    vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6[v_Index] := fl_IP_UdpServer_Start(v_ServerIpAddrv6, omit, tsc_PortNumber_DNS_Server, p_DrbInfo, p_PdnIndex);         // @sic R5-145754  sic@

    // ICMPv6

    vc_IP_PTC_PDN_AddrServer.ICMPv6[v_Index] := valueof(cs_IP_AddrServer_Init(v_ServerIpAddrv6, v_ClientIpAddrv6, p_PdnIndex));

    fl_Icmpv6Server_Start(cs_IPv6Addr(v_ServerIpAddrv6), p_DrbInfo);

    fl_Icmpv6Server_Start(cs_IPv6Addr(fl_IPv6AddrGetLinkLocalAddr(v_ServerIpAddrv6)), p_DrbInfo);

  }

	  altstep a_DNS_Handler() runs on IP_PTC

  { /* NOTE: handling of DNS is supported for UDP only */

    /* @sic R5-145754: support of DNS sic@ */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var IP_Socket_Type v_RemoteSocket;

    var IP_Socket_Type v_LocalSocket;

    var Datagram_UL_Type v_Datagram;

    var octetstring v_RespData;

    var IP_AddrServerArray_Type v_IP_AddrServerArray;

    var AddrServerInfo_Type v_AddrServerInfo;

    var IP_DrbInfo_Type v_DrbInfo;

    var PDN_Index_Type v_PdnIndex;

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(?, tsc_PortNumber_DNS_Server)))

      -> value  v_IP_SOCKET_IND

      {

        v_RemoteSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Remote;

        v_LocalSocket := v_IP_SOCKET_IND.DATA.ConnectionId.Local;

        v_Datagram := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom;

        if (ischosen(v_LocalSocket.IpAddr.V6)) {

          v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv6;

        } else {

          v_IP_AddrServerArray := vc_IP_PTC_PDN_AddrServer.DNS_UdpIPv4;

        }

        v_AddrServerInfo := valueof(fl_IP_AddrServer_Get(v_IP_AddrServerArray, v_LocalSocket.IpAddr));


        v_DrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_AddrServerInfo.PdnIndex);

        v_PdnIndex := v_AddrServerInfo.PdnIndex;

        v_RespData := f_DNS_GetEncodedResponse(v_Datagram.Buffer, v_PdnIndex);

        if (lengthof(v_RespData) >  0) {

          fl_IP_AddrServer_SendTriggerToGeran(v_DrbInfo);               // @sic R5-133612: trigger to GERAN sic@

          IP_SOCK.send(cs_UDP_DATA_REQ(v_LocalSocket, v_RemoteSocket, v_RespData));

        }

      }

  }


Change 5 – Coordination messages 
Change 5.1 – Type definitions
	Type/function name
	IP_ChangeDrbInfo_Type (new),
IP_ConfigReq_Type

	Reason for change
	There shall be distinct commands for the different use cases of routing table modifications

	Summary of changes
	new commands 'ChangeDrbInfo' and 'ChangeEutraCell' instead of 'ChangeCell'

	TTCN module
	IP_PTC_CtrlMsgs


	  type record IP_ChangeDrbInfo_Type {

    PDN_Index_Type      PdnId,

    IP_DrbInfo_Type     DrbInfo

  };

	  type union IP_ConfigReq_Type {        /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */

    IP_Start_Type                       StartIP,

    IP_Stop_Type                        StopIP,           /* @sic R5s131000: Null_Type replaced by IP_Stop_Type sic@ */

    Null_Type                           GeranCnf,         /* e.g. to confirm GeranTrigger */


    IP_ChangeDrbInfo_Type               ChangeDrbInfo,

    EUTRA_CellId_Type                   ChangeEutraCell,
    IP_SendRouterAdvertisement_Type     SendRouterAdvertisement,

    IP_RoutingInfo_Type                 ExplicitRouting,  /* @sic R5s130681 change 3: handling of RTP data sic@ */

    Null_Type                           ChangeToRawmode   /* @sic R5s150074 - Additional change sic@ */

  };


Change 5.2 – Templates
	Type/function name
	cs_IP_ChangeEutraCellReq,
cs_IP_ChangeDrbInfoReq (new),
cs_IP_ChangeUtranCellReq (removed),
cs_IP_ChangeGeranCellReq (removed)

	Reason for change
	new commads (see change 5.1)

	Summary of changes
	templates added/changed to cope with new commands

	TTCN module
	IP_PTC_CtrlMsgs


	  template (value) IP_ConfigReq_Type cs_IP_ChangeEutraCellReq(EUTRA_CellId_Type p_EutraCellId
) :=

  { /* @sic R5-125676: new parameter p_DrbId sic@ */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS) */

    ChangeEutraCell := p_EutraCellId






  };

	  template (value) IP_ConfigReq_Type cs_IP_ChangeDrbInfoReq(PDN_Index_Type p_PdnId,

                                                            template (value) IP_DrbInfo_Type p_DrbInfo) :=

  {

    ChangeDrbInfo := {

      PdnId := p_PdnId,

      DrbInfo := p_DrbInfo

    }

  };

	










	











Change 5.3 – Functions
	Type/function name
	f_IP_ChangeEutraCell,
f_IP_ChangeDrbMapping (new),
f_IP_ChangeDrbMappingEUTRA (new),
f_IP_ChangeDrbMappingUTRAN (replacing f_IP_ChangeUtranCell)

	Reason for change
	new commads (see change 5.1)

	Summary of changes
	functions added/changed to cope with new commands

	TTCN module
	IP_PTC_CtrlMsgs


	  function f_IP_ChangeEutraCell(IP_RAT_CTRL_PORT p_Port,

                                EUTRA_CellId_Type p_EutraCellId
)

  { /* @sic R5-125676: new parameter p_DrbId sic@ */

    p_Port.send(cs_IP_ChangeEutraCellReq(p_EutraCellId));

    p_Port.receive(cr_IP_CommonCnf);

  }

	  function f_IP_ChangeDrbMapping(IP_RAT_CTRL_PORT p_Port,

                                 template (value) IP_DrbInfo_Type p_DrbInfo,

                                 PDN_Index_Type p_PdnIndex := PDN_1)

  {

    p_Port.send(cs_IP_ChangeDrbInfoReq(p_PdnIndex, p_DrbInfo));

    p_Port.receive(cr_IP_CommonCnf);

  }

	  function f_IP_ChangeDrbMappingEUTRA(IP_RAT_CTRL_PORT p_Port,

                                      EUTRA_CellId_Type p_EutraCellId,

                                      IP_DrbId_Type p_DrbId,

                                      PDN_Index_Type p_PdnIndex := PDN_1)

  {

    f_IP_ChangeDrbMapping(p_Port, cs_DrbInfo_EUTRA(p_EutraCellId, p_DrbId), p_PdnIndex);

  }

	






  function f_IP_ChangeDrbMappingUTRAN(IP_RAT_CTRL_PORT p_Port,

                                      UTRAN_CellId_Type p_UtranCellId,

                                      IP_DrbId_Type p_DrbId)

  {

    f_IP_ChangeDrbMapping(p_Port, cs_DrbInfo_UTRAN(p_UtranCellId, p_DrbId));

  }


Change 5.4 – Further changes
	Type/function name
	f_UTRAN_IP_Handling_Change_Configuration, f_EUTRA_UE_Detach_SwitchOffUe_Trigger

	Reason for change
	1. f_IP_ChangeUtranCell has been renamed to f_IP_ChangeDrbMappingUTRAN
2. In EUTRA postamble there is no intra-EUTRA cell change but PDN1 shall be forced to be mapped to DRB1 for the given cell to serve all possible scenarios

	Summary of changes
	1. f_IP_ChangeUtranCell renamed to f_IP_ChangeDrbMappingUTRAN in f_UTRAN_IP_Handling_Change_Configuration
2. in postamble f_IP_ChangeDrbMappingEUTRA is used instead of f_IP_ChangeEutraCell

	TTCN module
	UTRAN_RAB_Functions, EUTRA_NASSteps


	  function f_UTRAN_IP_Handling_Change_Configuration(UTRAN_CellId_Type p_CellId,

                                                    RB_ConfigType p_RB_Configuration) runs on UTRAN_PTC

  {

    var integer v_DrbId := fl_UTRAN_IP_Handling_GetDrbId (p_RB_Configuration);  //@sic R5s140161 sic@

    f_IP_ChangeDrbMappingUTRAN (IP, p_CellId, v_DrbId);

  }

	  function f_EUTRA_UE_Detach_SwitchOffUe_Trigger(EUTRA_CellId_Type p_CellId,

                                                 template (omit) EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type p_UtCmd := omit,

                                                 boolean p_USIM_Rmvd := false) //@sic R5s140906 sic@

    runs on EUTRA_PTC return EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type

  {

    var EUTRA_UE_Detach_SwitchOffUe_UtCmd_Type v_UtCmd;








    f_IP_ChangeDrbMappingEUTRA(IP, p_CellId, tsc_DRB1, PDN_1);
    if (isvalue(p_UtCmd)) {

      v_UtCmd := valueof(p_UtCmd);

    }

    else if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) { // The UE can be switched off @sic R5s09225 sic@ or USIM removed @sic R5s120597r1 sic@

      v_UtCmd := switchOff;

    }

    else {

      v_UtCmd := powerOff

        }

    select (v_UtCmd) {

      case (powerOff) {

        f_UT_PowerOffUE(UT);

      }

      case (switchOff) {

        f_UT_SwitchOffUE(UT, p_USIM_Rmvd); // @sic R5s100584 sic@

      }

      case (detach) {

        f_UT_InitiateDetach(UT);

      }

      case else {

        FatalError(__FILE__, __LINE__, "");

      }

    }

    return v_UtCmd;

  }


Change 6 – IP PTC Main loop: ApplicationCtrl
	Type/function name
	fl_StopPDN, a_ApplicationCtrl

	Reason for change
	1. PDN mapping table needs to be updated when starting or stopping a PDN
2. New commands ChangeDrbInfo and ChangeEutraCell instead of ChangeCell

	Summary of changes
	1. PDN mapping table modified for StartIP and StopIP
2. Handling for ChangeDrbInfo and ChangeEutraCell added, removed for ChangeCell

	TTCN module
	IP_PTC_Main


	  function fl_StopPDN(integer p_Index) runs on IP_PTC

  {

    fl_IP_AddrServers_Stop(p_Index);

    vc_IP_PTC_IMS_Info.Server[p_Index] := f_IP_IMS_Stop(vc_IP_PTC_IMS_Info.Server[p_Index]); 
    vc_IP_PTC_PDN_MappingTable[p_Index].DefaultDrb := omit;
  }

	  altstep a_ApplicationCtrl(IP_PTC_CTRL_PORT p_CtrlPort) runs on IP_PTC

  {

    var IP_ConfigReq_Type v_IP_ConfigReq;

    var PDN_Index_Type v_PdnIndex;

    var IP_DrbInfo_Type v_DrbInfo;

    var IP_DrbInfo_Type v_CurrentDrbInfo;

    var EUTRA_CellId_Type v_EutraCellId;
    var integer i;

    [] p_CtrlPort.receive(cr_IP_CtrlAnyReq) -> value v_IP_ConfigReq {

      if (ischosen(v_IP_ConfigReq.StartIP)) {

        v_PdnIndex := v_IP_ConfigReq.StartIP.PdnId;

        v_DrbInfo := v_IP_ConfigReq.StartIP.DrbInfo;

        f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);
        fl_IP_AddrServers_Start(v_PdnIndex, v_DrbInfo);

        fl_IP_ImsServer_Start(v_PdnIndex);

      } else if (ischosen(v_IP_ConfigReq.StopIP)) {

        if (ischosen(v_IP_ConfigReq.StopIP.PdnId)) {       /* @sic R5s131000: stop particular PDN sic@ */

          i := f_PDN_PdnIndex2Integer(v_IP_ConfigReq.StopIP.PdnId);

          fl_StopPDN(i);

        } else {

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            fl_StopPDN(i);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeEutraCell)) {

        v_EutraCellId := v_IP_ConfigReq.ChangeEutraCell;

        if (f_RoutingTable_ChangeEutraCell(vc_IP_PTC_DRB_RoutingTable, v_EutraCellId)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          for (i:=0; i<tsc_NoOfPDNs; i:=i+1) {

            f_IP_DrbInfo_ChangeEutraCell(vc_IP_PTC_PDN_MappingTable[i].DefaultDrb, v_EutraCellId);

          }

        }

      } else if (ischosen(v_IP_ConfigReq.ChangeDrbInfo)) {

        v_PdnIndex := v_IP_ConfigReq.ChangeDrbInfo.PdnId;

        v_DrbInfo := v_IP_ConfigReq.ChangeDrbInfo.DrbInfo;

        v_CurrentDrbInfo := f_IP_PTC_PDN_GetDrbInfoForDefaultDrb(v_PdnIndex);

        if (f_RoutingTable_ChangeDrbInfo(vc_IP_PTC_DRB_RoutingTable, v_CurrentDrbInfo, v_DrbInfo)) {

          // we only change the routing at the SS when really needed

          f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

          f_IP_PTC_PDN_SetDrbInfoForDefaultDrb(v_PdnIndex, v_DrbInfo);

        }








      } else if (ischosen(v_IP_ConfigReq.SendRouterAdvertisement)) {

        fl_ICMPv6_SendRouterAdvertisement(v_IP_ConfigReq.SendRouterAdvertisement.PdnId);

      } else if (ischosen(v_IP_ConfigReq.ExplicitRouting)) {

        f_IP_PTC_RoutingTable_AddEntry(v_IP_ConfigReq.ExplicitRouting.IpInfo, v_IP_ConfigReq.ExplicitRouting.DRB, v_IP_ConfigReq.ExplicitRouting.Discard);

      } else if (ischosen(v_IP_ConfigReq.ChangeToRawmode)) {  /* @sic R5s150074 - Additional change sic@ */

        vc_IP_PTC_DRB_RoutingTable := {};

        f_IP_PTC_SendRoutingTable(vc_IP_PTC_DRB_RoutingTable);

      }

      p_CtrlPort.send(cs_IP_CommonCnf);

    }

  }


